// File: NFT2PRINT_LL.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.4;

import "@openzeppelin/contracts/access/AccessControl.sol";
import "@openzeppelin/contracts/security/ReentrancyGuard.sol";
import "@openzeppelin/contracts/token/ERC20/IERC20.sol";

//    _   _ _____ _____ ____  ____  ____  ___ _   _ _____
//   | \ | |  ___|_   _|___ \|  _ \|  _ \|_ _| \ | |_   _|
//   |  \| | |_    | |   __) | |_) | |_) || ||  \| | | |
//   | |\  |  _|   | |  / __/|  __/|  _ < | || |\  | | |
//   |_| \_|_|     |_| |_____|_|   |_| \_\___|_| \_| |_|
//

contract ToyNFTContract {
    function ownerOf(uint256 tokenId) public view virtual returns (address) {}
}

contract NFT2PrintLegendaryPass {
    function ownerOf(uint256 tokenId) public view virtual returns (address) {}
}

contract NFT2PRINT_LL is AccessControl, ReentrancyGuard {
    bytes32 public constant REGISTER_ROLE = keccak256("REGISTER_ROLE");

    struct ToyID {
        uint256 data;
        address toyCreatorAddres;
    }

    struct TokenInfo{
        IERC20 paytoken;
        uint256 priceStandard;
        uint256 priceLarge;
    }

    mapping (uint256 => TokenInfo) public AllowedCrypto;
    mapping (uint256 => ToyID) public toyData;

    bool public isProductionEnabled = true;
    uint256 public standard_ToyPrice = 0.08 ether;
    uint256 public large_ToyPrice = 0.4 ether;
    uint256 public legendary_Discount = 40;
    uint256 public totalToysCreated = 0;
    uint256 public maxLegendaryUsage = 5;

    mapping(uint256 => uint256) public legendaryID;

    ToyNFTContract private immutable toyaddr;
    NFT2PrintLegendaryPass private immutable legendPass;

    constructor(address ToyAddress, address PassAddress) {
        toyaddr = ToyNFTContract(ToyAddress);
        legendPass = NFT2PrintLegendaryPass(PassAddress);
        _setupRole(DEFAULT_ADMIN_ROLE, msg.sender);
    }

    function toggleToyProductionEnabled() external  onlyRole(DEFAULT_ADMIN_ROLE)
    {
        isProductionEnabled = !isProductionEnabled;
    }

    function setProductionPrice(uint256 _standardPrice, uint256 _largePrice) external onlyRole(DEFAULT_ADMIN_ROLE)   {
        standard_ToyPrice = _standardPrice;
        large_ToyPrice = _largePrice;
    }

    function setLegendaryDiscount(uint256 _discount) external onlyRole(DEFAULT_ADMIN_ROLE) {
        legendary_Discount = _discount;
    }

    function setLegendaryMaxUsage(uint256 _usage) external onlyRole(DEFAULT_ADMIN_ROLE) {
        maxLegendaryUsage = _usage;
    }

    function addPayToken (uint256 _pid, IERC20 _paytoken, uint256 _priceStandard, uint256 _priceLarge) public onlyRole(DEFAULT_ADMIN_ROLE) {
        
            TokenInfo storage token = AllowedCrypto[_pid];
                token.paytoken = _paytoken;
                token.priceStandard = _priceStandard;
                token.priceLarge = _priceLarge;
     }

    function createToy(uint256[] memory _tokenIdsStandard, uint256[] memory _tokenIdsLarge) public payable {
         
        uint256 totaltokenIdsStd = _tokenIdsStandard.length;
        uint256 totaltokenIdsLrg = _tokenIdsLarge.length;
        require(isProductionEnabled, "Production is disabled");
        require(msg.value >= (standard_ToyPrice * totaltokenIdsStd) + (large_ToyPrice * totaltokenIdsLrg) , "Ether value sent is not correct");

        for (uint256 i = 0; i < totaltokenIdsStd; i++) {
            require(toyaddr.ownerOf(_tokenIdsStandard[i]) == msg.sender, "You must own the orignal NFT");
        }

        for (uint256 i = 0; i < totaltokenIdsLrg; i++) {
            require(toyaddr.ownerOf(_tokenIdsLarge[i]) == msg.sender, "You must own the orignal NFT");
        }

        totalToysCreated += (totaltokenIdsStd + totaltokenIdsLrg) ;
    }

    function createToy_ERC20(uint256[] memory _tokenIdsStandard, uint256[] memory _tokenIdsLarge, uint256 _pid) public nonReentrant {
        
        TokenInfo storage tokens = AllowedCrypto[_pid];
        IERC20 paytoken;
        paytoken = tokens.paytoken;
        
        
        uint256 totaltokenIdsStd = _tokenIdsStandard.length;
        uint256 totaltokenIdsLrg = _tokenIdsLarge.length;
       
        require(isProductionEnabled, "Production is disabled");
        
        for (uint256 i = 0; i < totaltokenIdsStd; i++) {
            require(toyaddr.ownerOf(_tokenIdsStandard[i]) == msg.sender, "You must own the orignal NFT");
        }

        for (uint256 i = 0; i < totaltokenIdsLrg; i++) {
            require(toyaddr.ownerOf(_tokenIdsLarge[i]) == msg.sender, "You must own the orignal NFT");
        }

        require(paytoken.transferFrom(msg.sender, address(this), (tokens.priceStandard * totaltokenIdsStd) + (tokens.priceLarge * totaltokenIdsLrg)));
        
        totalToysCreated += totaltokenIdsStd + totaltokenIdsLrg;
    }

    function createToyLP_ERC20(uint256[] memory _tokenIdsStandard, uint256[] memory _tokenIdsLarge, uint256 _pid, uint256 _legendaryPassId) public nonReentrant {
        
        TokenInfo storage tokens = AllowedCrypto[_pid];
        IERC20 paytoken;
        paytoken = tokens.paytoken;
        
        uint256 totaltokenIdsStd = _tokenIdsStandard.length;
        uint256 totaltokenIdsLrg = _tokenIdsLarge.length;
        uint256 priceStandard = (((100 - legendary_Discount) * tokens.priceStandard) / 100);
        uint256 priceLarge = (((100 - legendary_Discount) * tokens.priceLarge) / 100);

        require(isProductionEnabled, "Production is disabled");
        require(legendPass.ownerOf(_legendaryPassId) == msg.sender, "You must own the NFT2Print Legendary Pass");
        require(legendaryID[_legendaryPassId] + totaltokenIdsStd + totaltokenIdsLrg <= maxLegendaryUsage, "Out of Legendary pass usage");

        for (uint256 i = 0; i < totaltokenIdsStd; i++) {
            require(toyaddr.ownerOf(_tokenIdsStandard[i]) == msg.sender, "You must own the orignal NFT");
        }

        for (uint256 i = 0; i < totaltokenIdsLrg; i++) {
            require(toyaddr.ownerOf(_tokenIdsLarge[i]) == msg.sender, "You must own the orignal NFT");
        }

        require(paytoken.transferFrom(msg.sender, address(this), (priceStandard * totaltokenIdsStd) + (priceLarge * totaltokenIdsLrg)));

        totalToysCreated += totaltokenIdsStd + totaltokenIdsLrg;
        legendaryID[_legendaryPassId] += totaltokenIdsStd + totaltokenIdsLrg;
    }
    
    
    function createToyLP(uint256[] memory _tokenIdsStandard, uint256[] memory _tokenIdsLarge, uint256 _legendaryPassId) public payable nonReentrant
    {
        uint256 totaltokenIdsStd = _tokenIdsStandard.length;
        uint256 totaltokenIdsLrg = _tokenIdsLarge.length;
        uint256 priceStandard = (((100 - legendary_Discount) * standard_ToyPrice) / 100);
        uint256 priceLarge = (((100 - legendary_Discount) * large_ToyPrice) / 100);
        
        require(isProductionEnabled, "Production is disabled");
        require(legendPass.ownerOf(_legendaryPassId) == msg.sender, "You must own the NFT2Print Legendary Pass");
        require(legendaryID[_legendaryPassId] + totaltokenIdsStd + totaltokenIdsLrg <= maxLegendaryUsage, "Out of Legendary pass usage");
        require(msg.value >= (priceStandard * totaltokenIdsStd) + (priceLarge * totaltokenIdsLrg), "Ether value sent is not correct");

        for (uint256 i = 0; i < totaltokenIdsStd; i++) {
            require(toyaddr.ownerOf(_tokenIdsStandard[i]) == msg.sender, "You must own the orignal NFT");
        }

        for (uint256 i = 0; i < totaltokenIdsLrg; i++) {
            require(toyaddr.ownerOf(_tokenIdsLarge[i]) == msg.sender, "You must own the orignal NFT");
        }

        totalToysCreated += totaltokenIdsStd + totaltokenIdsLrg;
        legendaryID[_legendaryPassId] += totaltokenIdsStd + totaltokenIdsLrg;
    }

    function registerToy(uint256 _uidToy, uint256 _data, address _toyCreatorAddress) external onlyRole(REGISTER_ROLE) {
        ToyID storage toy = toyData[_uidToy];

        toy.data = _data;
        toy.toyCreatorAddres = _toyCreatorAddress;
    }

    function registerToyBatch(uint256[] memory _uidToy, uint256[] memory _data, address[] memory _toyCreatorAddress
    ) external onlyRole(REGISTER_ROLE) {
        uint256 uidLenght = _uidToy.length;
        uint256 dataLenght = _data.length;
        uint256 addressLenght = _toyCreatorAddress.length;

        require((uidLenght == dataLenght) && (addressLenght == dataLenght), "Arrays not the same size");

        for (uint256 i = 0; i < uidLenght; i++) {
           
            ToyID storage toy = toyData[_uidToy[i]];
            toy.data = _data[i];
            toy.toyCreatorAddres = _toyCreatorAddress[i];
        }
    }

     function getERC20_token(uint256 _pid) public view virtual returns(IERC20, uint256, uint256) {
            TokenInfo storage token = AllowedCrypto[_pid];
            IERC20 paytoken;
            paytoken = token.paytoken;
            uint256 priceStandard = token.priceStandard;
            uint256 priceLarge = token.priceLarge;
            return (paytoken, priceStandard, priceLarge);
        }

    function getLegendaryPassUsages(uint256 _tokenID) public view returns (uint256) {
        return legendaryID[_tokenID];
    }


    function withdraw() external  onlyRole(DEFAULT_ADMIN_ROLE) {
        uint256 contractBalance = address(this).balance;
        bool success = true;

        (success, ) = payable(0x0108E1348D42192a749Ae029fa2Ad7b5c2E6F210).call{
            value: (95 * contractBalance) / 100
        }("");
        require(success, "Transfer failed");

        (success, ) = payable(0x4B8229Db7bBd5901FfbB387A0E7c67E3fb90fC61).call{
            value: (5 * contractBalance) / 100
        }("");
        require(success, "Transfer failed");
    }

    function withdrawERC20(uint256 _pid) external onlyRole(DEFAULT_ADMIN_ROLE) {
        TokenInfo storage tokens = AllowedCrypto[_pid];
        IERC20 paytoken;
        paytoken = tokens.paytoken;
        paytoken.transfer(msg.sender, paytoken.balanceOf(address(this)));
    }

}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) external returns (bool);
}

// File: @openzeppelin/contracts/security/ReentrancyGuard.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (security/ReentrancyGuard.sol)

pragma solidity ^0.8.0;

/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuard {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant _NOT_ENTERED = 1;
    uint256 private constant _ENTERED = 2;

    uint256 private _status;

    constructor() {
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and making it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        // On the first call to nonReentrant, _notEntered will be true
        require(_status != _ENTERED, "ReentrancyGuard: reentrant call");

        // Any calls to nonReentrant after this point will fail
        _status = _ENTERED;

        _;

        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = _NOT_ENTERED;
    }
}

// File: @openzeppelin/contracts/access/AccessControl.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/AccessControl.sol)

pragma solidity ^0.8.0;

import "./IAccessControl.sol";
import "../utils/Context.sol";
import "../utils/Strings.sol";
import "../utils/introspection/ERC165.sol";

/**
 * @dev Contract module that allows children to implement role-based access
 * control mechanisms. This is a lightweight version that doesn't allow enumerating role
 * members except through off-chain means by accessing the contract event logs. Some
 * applications may benefit from on-chain enumerability, for those cases see
 * {AccessControlEnumerable}.
 *
 * Roles are referred to by their `bytes32` identifier. These should be exposed
 * in the external API and be unique. The best way to achieve this is by
 * using `public constant` hash digests:
 *
 * ```
 * bytes32 public constant MY_ROLE = keccak256("MY_ROLE");
 * ```
 *
 * Roles can be used to represent a set of permissions. To restrict access to a
 * function call, use {hasRole}:
 *
 * ```
 * function foo() public {
 *     require(hasRole(MY_ROLE, msg.sender));
 *     ...
 * }
 * ```
 *
 * Roles can be granted and revoked dynamically via the {grantRole} and
 * {revokeRole} functions. Each role has an associated admin role, and only
 * accounts that have a role's admin role can call {grantRole} and {revokeRole}.
 *
 * By default, the admin role for all roles is `DEFAULT_ADMIN_ROLE`, which means
 * that only accounts with this role will be able to grant or revoke other
 * roles. More complex role relationships can be created by using
 * {_setRoleAdmin}.
 *
 * WARNING: The `DEFAULT_ADMIN_ROLE` is also its own admin: it has permission to
 * grant and revoke this role. Extra precautions should be taken to secure
 * accounts that have been granted it.
 */
abstract contract AccessControl is Context, IAccessControl, ERC165 {
    struct RoleData {
        mapping(address => bool) members;
        bytes32 adminRole;
    }

    mapping(bytes32 => RoleData) private _roles;

    bytes32 public constant DEFAULT_ADMIN_ROLE = 0x00;

    /**
     * @dev Modifier that checks that an account has a specific role. Reverts
     * with a standardized message including the required role.
     *
     * The format of the revert reason is given by the following regular expression:
     *
     *  /^AccessControl: account (0x[0-9a-f]{40}) is missing role (0x[0-9a-f]{64})$/
     *
     * _Available since v4.1._
     */
    modifier onlyRole(bytes32 role) {
        _checkRole(role);
        _;
    }

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IAccessControl).interfaceId || super.supportsInterface(interfaceId);
    }

    /**
     * @dev Returns `true` if `account` has been granted `role`.
     */
    function hasRole(bytes32 role, address account) public view virtual override returns (bool) {
        return _roles[role].members[account];
    }

    /**
     * @dev Revert with a standard message if `_msgSender()` is missing `role`.
     * Overriding this function changes the behavior of the {onlyRole} modifier.
     *
     * Format of the revert message is described in {_checkRole}.
     *
     * _Available since v4.6._
     */
    function _checkRole(bytes32 role) internal view virtual {
        _checkRole(role, _msgSender());
    }

    /**
     * @dev Revert with a standard message if `account` is missing `role`.
     *
     * The format of the revert reason is given by the following regular expression:
     *
     *  /^AccessControl: account (0x[0-9a-f]{40}) is missing role (0x[0-9a-f]{64})$/
     */
    function _checkRole(bytes32 role, address account) internal view virtual {
        if (!hasRole(role, account)) {
            revert(
                string(
                    abi.encodePacked(
                        "AccessControl: account ",
                        Strings.toHexString(uint160(account), 20),
                        " is missing role ",
                        Strings.toHexString(uint256(role), 32)
                    )
                )
            );
        }
    }

    /**
     * @dev Returns the admin role that controls `role`. See {grantRole} and
     * {revokeRole}.
     *
     * To change a role's admin, use {_setRoleAdmin}.
     */
    function getRoleAdmin(bytes32 role) public view virtual override returns (bytes32) {
        return _roles[role].adminRole;
    }

    /**
     * @dev Grants `role` to `account`.
     *
     * If `account` had not been already granted `role`, emits a {RoleGranted}
     * event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     *
     * May emit a {RoleGranted} event.
     */
    function grantRole(bytes32 role, address account) public virtual override onlyRole(getRoleAdmin(role)) {
        _grantRole(role, account);
    }

    /**
     * @dev Revokes `role` from `account`.
     *
     * If `account` had been granted `role`, emits a {RoleRevoked} event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     *
     * May emit a {RoleRevoked} event.
     */
    function revokeRole(bytes32 role, address account) public virtual override onlyRole(getRoleAdmin(role)) {
        _revokeRole(role, account);
    }

    /**
     * @dev Revokes `role` from the calling account.
     *
     * Roles are often managed via {grantRole} and {revokeRole}: this function's
     * purpose is to provide a mechanism for accounts to lose their privileges
     * if they are compromised (such as when a trusted device is misplaced).
     *
     * If the calling account had been revoked `role`, emits a {RoleRevoked}
     * event.
     *
     * Requirements:
     *
     * - the caller must be `account`.
     *
     * May emit a {RoleRevoked} event.
     */
    function renounceRole(bytes32 role, address account) public virtual override {
        require(account == _msgSender(), "AccessControl: can only renounce roles for self");

        _revokeRole(role, account);
    }

    /**
     * @dev Grants `role` to `account`.
     *
     * If `account` had not been already granted `role`, emits a {RoleGranted}
     * event. Note that unlike {grantRole}, this function doesn't perform any
     * checks on the calling account.
     *
     * May emit a {RoleGranted} event.
     *
     * [WARNING]
     * ====
     * This function should only be called from the constructor when setting
     * up the initial roles for the system.
     *
     * Using this function in any other way is effectively circumventing the admin
     * system imposed by {AccessControl}.
     * ====
     *
     * NOTE: This function is deprecated in favor of {_grantRole}.
     */
    function _setupRole(bytes32 role, address account) internal virtual {
        _grantRole(role, account);
    }

    /**
     * @dev Sets `adminRole` as ``role``'s admin role.
     *
     * Emits a {RoleAdminChanged} event.
     */
    function _setRoleAdmin(bytes32 role, bytes32 adminRole) internal virtual {
        bytes32 previousAdminRole = getRoleAdmin(role);
        _roles[role].adminRole = adminRole;
        emit RoleAdminChanged(role, previousAdminRole, adminRole);
    }

    /**
     * @dev Grants `role` to `account`.
     *
     * Internal function without access restriction.
     *
     * May emit a {RoleGranted} event.
     */
    function _grantRole(bytes32 role, address account) internal virtual {
        if (!hasRole(role, account)) {
            _roles[role].members[account] = true;
            emit RoleGranted(role, account, _msgSender());
        }
    }

    /**
     * @dev Revokes `role` from `account`.
     *
     * Internal function without access restriction.
     *
     * May emit a {RoleRevoked} event.
     */
    function _revokeRole(bytes32 role, address account) internal virtual {
        if (hasRole(role, account)) {
            _roles[role].members[account] = false;
            emit RoleRevoked(role, account, _msgSender());
        }
    }
}

// File: @openzeppelin/contracts/utils/introspection/ERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/ERC165.sol)

pragma solidity ^0.8.0;

import "./IERC165.sol";

/**
 * @dev Implementation of the {IERC165} interface.
 *
 * Contracts that want to implement ERC165 should inherit from this contract and override {supportsInterface} to check
 * for the additional interface id that will be supported. For example:
 *
 * ```solidity
 * function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
 *     return interfaceId == type(MyInterface).interfaceId || super.supportsInterface(interfaceId);
 * }
 * ```
 *
 * Alternatively, {ERC165Storage} provides an easier to use but more expensive implementation.
 */
abstract contract ERC165 is IERC165 {
    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IERC165).interfaceId;
    }
}

// File: @openzeppelin/contracts/utils/Strings.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (utils/Strings.sol)

pragma solidity ^0.8.0;

/**
 * @dev String operations.
 */
library Strings {
    bytes16 private constant _HEX_SYMBOLS = "0123456789abcdef";
    uint8 private constant _ADDRESS_LENGTH = 20;

    /**
     * @dev Converts a `uint256` to its ASCII `string` decimal representation.
     */
    function toString(uint256 value) internal pure returns (string memory) {
        // Inspired by OraclizeAPI's implementation - MIT licence
        // https://github.com/oraclize/ethereum-api/blob/b42146b063c7d6ee1358846c198246239e9360e8/oraclizeAPI_0.4.25.sol

        if (value == 0) {
            return "0";
        }
        uint256 temp = value;
        uint256 digits;
        while (temp != 0) {
            digits++;
            temp /= 10;
        }
        bytes memory buffer = new bytes(digits);
        while (value != 0) {
            digits -= 1;
            buffer[digits] = bytes1(uint8(48 + uint256(value % 10)));
            value /= 10;
        }
        return string(buffer);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
     */
    function toHexString(uint256 value) internal pure returns (string memory) {
        if (value == 0) {
            return "0x00";
        }
        uint256 temp = value;
        uint256 length = 0;
        while (temp != 0) {
            length++;
            temp >>= 8;
        }
        return toHexString(value, length);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
     */
    function toHexString(uint256 value, uint256 length) internal pure returns (string memory) {
        bytes memory buffer = new bytes(2 * length + 2);
        buffer[0] = "0";
        buffer[1] = "x";
        for (uint256 i = 2 * length + 1; i > 1; --i) {
            buffer[i] = _HEX_SYMBOLS[value & 0xf];
            value >>= 4;
        }
        require(value == 0, "Strings: hex length insufficient");
        return string(buffer);
    }

    /**
     * @dev Converts an `address` with fixed length of 20 bytes to its not checksummed ASCII `string` hexadecimal representation.
     */
    function toHexString(address addr) internal pure returns (string memory) {
        return toHexString(uint256(uint160(addr)), _ADDRESS_LENGTH);
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/access/IAccessControl.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (access/IAccessControl.sol)

pragma solidity ^0.8.0;

/**
 * @dev External interface of AccessControl declared to support ERC165 detection.
 */
interface IAccessControl {
    /**
     * @dev Emitted when `newAdminRole` is set as ``role``'s admin role, replacing `previousAdminRole`
     *
     * `DEFAULT_ADMIN_ROLE` is the starting admin for all roles, despite
     * {RoleAdminChanged} not being emitted signaling this.
     *
     * _Available since v3.1._
     */
    event RoleAdminChanged(bytes32 indexed role, bytes32 indexed previousAdminRole, bytes32 indexed newAdminRole);

    /**
     * @dev Emitted when `account` is granted `role`.
     *
     * `sender` is the account that originated the contract call, an admin role
     * bearer except when using {AccessControl-_setupRole}.
     */
    event RoleGranted(bytes32 indexed role, address indexed account, address indexed sender);

    /**
     * @dev Emitted when `account` is revoked `role`.
     *
     * `sender` is the account that originated the contract call:
     *   - if using `revokeRole`, it is the admin role bearer
     *   - if using `renounceRole`, it is the role bearer (i.e. `account`)
     */
    event RoleRevoked(bytes32 indexed role, address indexed account, address indexed sender);

    /**
     * @dev Returns `true` if `account` has been granted `role`.
     */
    function hasRole(bytes32 role, address account) external view returns (bool);

    /**
     * @dev Returns the admin role that controls `role`. See {grantRole} and
     * {revokeRole}.
     *
     * To change a role's admin, use {AccessControl-_setRoleAdmin}.
     */
    function getRoleAdmin(bytes32 role) external view returns (bytes32);

    /**
     * @dev Grants `role` to `account`.
     *
     * If `account` had not been already granted `role`, emits a {RoleGranted}
     * event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     */
    function grantRole(bytes32 role, address account) external;

    /**
     * @dev Revokes `role` from `account`.
     *
     * If `account` had been granted `role`, emits a {RoleRevoked} event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     */
    function revokeRole(bytes32 role, address account) external;

    /**
     * @dev Revokes `role` from the calling account.
     *
     * Roles are often managed via {grantRole} and {revokeRole}: this function's
     * purpose is to provide a mechanism for accounts to lose their privileges
     * if they are compromised (such as when a trusted device is misplaced).
     *
     * If the calling account had been granted `role`, emits a {RoleRevoked}
     * event.
     *
     * Requirements:
     *
     * - the caller must be `account`.
     */
    function renounceRole(bytes32 role, address account) external;
}

// File: @openzeppelin/contracts/utils/introspection/IERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/IERC165.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}